

# 共働き世帯と片働き世帯の収入・支出動向の比較

～子供の数と妻の働き方を考慮した分析～

応募部門：B クラス（SAS または JMP の使用歴 3 年未満）

チーム名：MOUNTAINS

中山 貴公、山川 雄也

(株式会社 浜銀総合研究所 情報戦略コンサルティング部)

Comparison of income and consumption trends of dual-income households and households with a fulltime homemaker

(Analysis considering the number of children and how spouses work)

Yoshimasa Nakayama, Yuya Yamakawa

Hamagin Research Institute, Ltd.

## 要旨

増加傾向にある共働き世帯の収入・支出両面の特徴をつかむことを目的とし、SAS ユーザー会世話人によって作成された「新疑似マイクロデータ」を用いて分析を行った。その際、便宜的ではあるが、共働き世帯において妻の収入が一定水準より低い世帯を「妻パート」世帯、一定水準より高い世帯を「妻フルタイム」世帯と切り分けることで、妻の就業形態を考慮できるようにした。

分析の結果、世帯全体の収入においては片働き世帯と妻パート世帯の間では大きな差はないが、フルタイム世帯との間では大きな差が見られた。また、子供の数が増えると共働き世帯内の妻パートの割合が上昇することが示された。支出面においては、子供の数が増えるほど、また妻の就業形態がフルタイムに近づくほど、食料や教育費にかかる支出が高まる。そのしわ寄せが預貯金純増分に現れており、特に子育て中の妻専業主婦世帯・妻パート世帯で預貯金が増やせていない世帯が多いことが示された。

キーワード：共働き世帯、消費支出、預貯金、新疑似マイクロデータ、平成 16 年全国消費実態調査

## 1. はじめに

近年、女性の社会進出が進み、共働き世帯が増加してきた(図表 1)。そうした中で、核家族世帯や DINKs (子供を持たない共働き夫婦) 世帯が増加し、日本社会における家族のあり方が変わってきた。世帯類型の変容の背景には女性の社会進出のみならず、家計における経済的な側面も影響しているのではないだろうか。

そこで本分析では、「平成 16 年全国消費実態調査」の匿名データをもとに SAS ユーザー会世話人によって作成された「新疑似マイクロデータ」を利用し、夫婦の働き方による収入や支出動向の違いから生活ぶりを比較し、特徴を見出すことを目的とする。その際、一口に共働き世帯と言っても、妻がパートタイマ

一であるか、フルタイム就労であるかによって、世帯全体の収入の水準やライフスタイルに差があると考えられる。しかし、新疑似マイクロデータには世帯主の配偶者の就業形態についての項目は含まれていない。そこで、共働き世帯の中でも妻の収入が一定水準（後述）より低い世帯を「妻パート世帯」、高い世帯を「妻フルタイム世帯」とみなすことで、妻の働き方を考慮した比較を行う。

## 2. 分析対象および抽出条件

分析対象は「夫婦のみ世帯」および「夫婦+子供（うち1人以上が未就学児）の二世代の世帯」とする。これは、できる限り収入面での条件をそろえた上で片働き世帯と共働き世帯を比較するためである。例えば、「二世代の世帯」で就業人員が2人であった場合、夫婦とその親夫婦との二世代の世帯であれば、夫婦の共働きではなく、夫婦と親夫婦のうち1人ずつが働いているというケースも考えられるためである。

このような背景を踏まえ、以下の図表2に挙げた条件全てを満たす世帯を分析対象として抽出する。二世代の世帯において、先に述べた例のように親世代との同居というケースを除くため、未就学児の有無を抽出条件に含めている。これにより、二世代の世帯において、世帯人員（X03）が3であれば「夫婦+子供1人世帯」、4であれば「夫婦+子供2人世帯」、5であれば「夫婦+子供3人以上世帯」という解釈が可能となる。一方で、就学後の子供しかいない世帯は分析対象から除いていることには留意する必要がある。

図表2 分析対象の抽出条件

項目	内容	条件式	
世帯区分	勤労者世帯に限定（※1）	X02 = 1	
世帯主の性別	世帯主が男性の場合に限定（※2）	X08 = 1	
世帯主の年齢	49歳以下に限定	X09 <= 10	
家族分類	夫婦のみの世帯・二世代の世帯に限定	X11 in (2, 3)	
未就学児の有無	（二世代の世帯に限り）「有り」に限定（※3）	X12 = 2	
公的年金給付	配偶者が60歳以上で年金を受給しているケースを除く（※4）	Y018 = 0	
就業人員	図表3の【片働き・共働きの判定条件】の判定結果と整合的なケースに限定	片働き世帯の場合	X04 = 1
		共働き世帯の場合	X04 = 2

（※1）「勤労者以外の世帯」は世帯主の月収が調査されていないため。


（※2）全国消費実態調査における「世帯主」は登記上のそれではなく、主たる稼ぎ手の意であるため。

（※3）二世代の世帯の家族構成を「世帯主夫婦+未婚の子」に限定するため。

（※4）経常的な収入の条件をなるべくそろえるため。

片働き・共働きの判定条件は図表3の通りである。抽出条件で世帯主を男性に固定しているため、世帯主を夫、その配偶者を妻と読み替えることができる。抽出条件に挙げた「就業人員」とこの判定条件が整合的でない場合（例えば「片働き世帯」と判定されたのに就業人員が2人であるケース）や、夫婦以外の世帯員の収入が発生しているケースが少数ではあるが存在するため、そうした世帯は上記の表で述べた通り、分析対象から除く。

図表1 専業主婦世帯と共働き世帯数の推移



また、先述したように「新疑似マイクロデータ」には配偶者の就業形態の情報は掲載されていない。そこで、本研究では便宜的に月ごとの妻の勤め先からの収入を12倍したものを妻の年収とみなし、それが配偶者特別控除の無くなるボーダーラインである141万円を超える世帯を「妻フルタイム」世帯、超えない世帯を「妻パート」世帯と判定する。なお、先に述べた抽出条件で世帯主の性別を男性に限定しているため、「配偶者の勤め先収入」が妻の勤め先からの収入を意味する。

図表3 分析対象内における就業形態の判定条件

	分類	判定基準	条件式
片働き・共働きの判定条件	片働き世帯	世帯主のみに勤め先収入がある(※1)	Y006 > 0, Y007 = 0, Y008 = 0
	共働き世帯	世帯主とその配偶者のみに勤め先収入がある(※1)	Y006 > 0, Y007 > 0, Y008 = 0
妻の就業形態の判定条件	妻パート	共働き世帯のうち、配偶者の勤め先収入が年間で141万円以下(※2)	Y007 * 12 <= 1410000
	妻フルタイム	共働き世帯のうち、配偶者の勤め先収入が年間で141万円超(※2)	Y007 * 12 > 1410000

(※1) 子供が働いているなど、夫婦以外の働き手がいるケースを除くべく、Y008(他の世帯員の勤め先収入) = 0を条件に加えている。

(※2) 本分析で境界の値として用いた141万円は配偶者特別控除が無くなる収入の水準(いわゆる「141万円の壁」)であることや、抽出した共働き世帯3,421件における妻の勤め先収入において、141万円以下である割合が約52.6%となり、2004年の労働力調査における共働き世帯に占める妻の労働時間が0~34時間である世帯の割合(55.4%)に近いことから選定した。

以上の条件により、「新疑似マイクロデータ」の69,131世帯から8,972件(乗率を用いない場合)が抽出される。妻の就業形態や世帯人数による内訳は図表4にある通りである。また、図表5からは、子供の数が増えると妻パートの割合が増え、妻フルタイムの割合が減少することが分かる。結婚時には共働きであったとしても出産を機に離職する世帯や、子育てが落ち着いてもフルタイムでの職場復帰はせず、パートのような短時間労働で家庭との両立を図ろうとする世帯が多い様子がうかがえる(この点については比較の内容や分析対象、分析データは異なるが、井上・矢田[1]と概ね同様の示唆となる)。


図表4 分析対象内の妻の就業形態別世帯分類および世帯人数別世帯数・構成比

世帯分類\世帯人数	世帯数				
	2人(夫婦のみ)	3人(夫婦+子1人)	4人(夫婦+子2人)	5人(夫婦+子3人)以上	計
01.片働き(妻専業主婦)	876	1,739	2,051	885	5,551
共働き世帯全体	1,198	718	990	515	3,421
02.共働き(妻パート)	476	350	611	363	1,800
03.共働き(妻フルタイム)	722	368	379	152	1,621
抽出対象全体	2,074	2,457	3,041	1,400	8,972


世帯分類\世帯人数	構成比(%)				
	2人(夫婦のみ)	3人(夫婦+子1人)	4人(夫婦+子2人)	5人(夫婦+子3人)以上	計
01.片働き(妻専業主婦)	42.2	70.8	67.4	63.2	61.9
共働き世帯全体	57.8	29.2	32.6	36.8	38.1
02.共働き(妻パート)	23.0	14.2	20.1	25.9	20.1
03.共働き(妻フルタイム)	34.8	15.0	12.5	10.9	18.1
抽出対象全体	100.0	100.0	100.0	100.0	100.0

図表5 分析対象内の世帯人数別に見た妻の就業形態割合の違い



抽出対象において、世帯人員ごとの年齢構成の分布(図表6)をみると、世帯人員3人以上では30代付近にボリュームゾーンがあり、世帯人員が多くなるほど世帯主の年齢が高い世帯が多くなっている。一方、世帯人員が2人(夫婦のみ世帯)の場合は40代の構成比も高い。夫婦のみ世帯の中に、子供がいない夫婦と子育てがすでに終わった夫婦が混在していることが要因と考えられる。

図表6 世帯人員別に見た年齢構成の分布



### 3. 収入面の分析・考察

世帯全体の収入（「年間収入（単位万円）」）は平均値・中央値ともに、妻フルタイム世帯と他の世帯分類との間で大きな差がみられた（図表7）。また、箱ひげ図（図表8）から分かる通り、妻フルタイム世帯は他の世帯分類と比べて「ひげ」が長く、他の世帯分類と比べて分布の裾が長いことが分かる。


一方、世帯主（夫）の勤め先収入は妻パート世帯が他の世帯分類と比べて低水準となっている。このことから妻パート世帯は夫の収入が足りない分を埋め合わせるために妻がパート労働に従事していると考えられる。

図表7 世帯分類ごとの平均年齢・年間収入・世帯主（夫）の勤め先収入

	世帯数	年齢	年間収入 (単位:万円)		世帯主の勤め先収入 (単位:円)	
		平均値	平均値	中央値	平均値	中央値
201_2人・片働き(妻専業主婦)	876	35.9	549	512	378,511	354,892
202_2人・共働き(妻パート)	476	37.1	610	568	331,709	308,328
203_2人・共働き(妻フルタイム)	722	35.2	790	747	383,666	352,135
301_3人・片働き(妻専業主婦)	1,739	33.8	522	493	379,318	357,974
302_3人・共働き(妻パート)	350	33.4	517	494	303,434	287,833
303_3人・共働き(妻フルタイム)	368	33.4	734	711	340,752	322,429
401_4人・片働き(妻専業主婦)	2,051	35.7	555	528	390,112	373,754
402_4人・共働き(妻パート)	611	35.4	558	513	340,813	320,181
403_4人・共働き(妻フルタイム)	379	37.2	794	732	376,462	360,466
501_5人以上・片働き(妻専業主婦)	885	36.8	582	543	414,020	391,127
502_5人以上・共働き(妻パート)	363	36.6	595	536	354,508	343,091
503_5人以上・共働き(妻フルタイム)	152	38.2	796	768	409,135	389,837
計	8,972	35.4	594	547	375,170	353,451

※平均年齢は符号表における各階級の中央値に置き換えて算出した。  
 (例えば、「25～29歳」は27歳とみなした。また、「24歳以下」は22歳とみなした。)

図表8 世帯分類ごとの世帯全体の年間収入の箱ひげ図




(注)「ひげ」は箱の上端・下端から四分位範囲×1.5までの範囲で最も離れた観測値まで伸びている。[2][3]  
 (注)この箱ひげ図では外れ値の出力を抑制している。

## 4. 支出面の分析・考察

### 4.1. 消費支出

世帯人員・妻の就業形態ごとに消費支出のうちの10大費目の構成比をみると、人数が増えるにしたがって、食料や教育に対する支出の消費支出全体に占める割合が増えることが分かる（図表9）。

図表9 世帯分類ごとにみた消費支出全体に占める10大費目の構成比



ただし、世帯人員や妻の就業形態によって消費支出の総額の水準に差がみられるため、構成比だけでなく、実額も考慮する必要がある。そこで、実額（平均値）を見ると、消費支出全体の規模は世帯人数が増えるほど増加し、同じ世帯人数の中では妻の就業形態によっても差がみられる（図表10）。また、構成比で見た際に、消費支出全体に占める割合が増えていた食料や教育に対する支出は、実額の水準においても世帯人数や妻の就業形態によって差がみられる。

図表10 世帯分類ごとにみた消費支出全体および10大費目への支出（実額）

	世帯数	世帯主 平均年齢	年間収入 (単位:万円)	消費支出(単位:円)										
				全体	食料	住居	光熱・水道	家具・ 家事用品	被服 及び履物	保健医療	交通・通信	教育	教養娯楽	その他の 消費支出
201.2人・片働き(妻専業主婦)	876	35.9	549	259,045	52,912	31,196	14,200	8,203	11,296	8,417	45,620	212	27,795	59,194
202.2人・共働き(妻パート)	476	37.1	610	282,785	56,425	30,580	13,690	7,265	13,035	9,740	44,981	0	31,171	75,887
203.2人・共働き(妻フルタイム)	722	35.2	790	327,177	62,341	30,773	14,220	8,313	17,699	8,380	58,005	193	38,825	88,427
301.3人・片働き(妻専業主婦)	1,739	33.8	522	270,297	54,316	32,320	15,514	9,115	13,287	12,830	46,558	7,711	26,893	51,754
302.3人・共働き(妻パート)	350	33.4	517	278,065	50,331	28,204	15,035	7,757	12,267	10,554	62,034	11,590	25,223	55,071
303.3人・共働き(妻フルタイム)	368	33.4	734	317,516	59,390	28,352	15,677	8,733	17,658	11,042	53,918	14,711	32,395	75,640
401.4人・片働き(妻専業主婦)	2,051	35.7	555	273,660	62,778	21,982	17,068	8,334	12,614	11,659	38,561	19,854	30,614	50,195
402.4人・共働き(妻パート)	611	35.4	558	284,949	63,582	23,138	17,558	6,810	12,635	10,284	41,887	25,914	28,442	54,698
403.4人・共働き(妻フルタイム)	379	37.2	794	341,417	70,067	18,105	18,260	8,885	16,840	11,548	61,698	28,579	35,665	71,772
501.5人以上・片働き(妻専業主婦)	885	36.8	582	299,971	73,691	20,872	19,163	9,260	13,377	12,286	37,872	28,243	34,221	50,986
502.5人以上・共働き(妻パート)	363	36.6	595	317,229	72,537	15,563	19,509	9,442	11,054	12,776	64,598	27,838	30,144	53,770
503.5人以上・共働き(妻フルタイム)	152	38.2	796	355,793	82,304	19,585	20,352	8,242	16,703	16,673	48,966	32,271	39,363	71,335

※平均値は小数点以下を丸めているため、「食料」～「その他の消費支出」の合計が消費支出全体と一致しない場合がある。

食料への支出について見ると、食料全体への支出は世帯人数が増えるほど増加するとともに、妻フルタイム世帯においては調理食品や一般外食への支出額が多い点が共通してみられる（図表11）。一方、世帯人員が増えると調理食品への支出が増え、一般外食への支出が減る傾向がみられる。外食を減らし、自宅での食事を増やすことで食費の抑制を図っていると考えられる。

また、消費支出内における住居への支出は、世帯人数が増えることで平均的に減少しているように見えるが、消費支出には住宅ローンの返済が含まれていないことに注意する必要がある。実支出以外の支出

に含まれる「土地家屋借金返済」が0円超である世帯の割合は、世帯人数が増えるほど大きくなる傾向がみられる。

図表 11 世帯分類ごとにみた食料および住居への支出（実額）の詳細

	世帯数	世帯主 平均年齢	年間収入 (単位:万円)	食料(単位:円)			住居(単位:円)		住宅ローンを含めた 住居費(単位:円)			土地家屋借 金返済>0 の世帯割合
				全体	調理食品	一般外食	全体	うち家賃地 代	土地家屋 借金返済	家賃地代+ 土地家屋借 金返済		
201.2人・片働き(妻専業主婦)	876	35.9	549	52,912	5,200	12,801	31,196	29,018	22,925	51,942	34.9%	
202.2人・共働き(妻パート)	476	37.1	610	56,425	6,368	15,910	30,580	29,769	22,531	52,299	37.6%	
203.2人・共働き(妻フルタイム)	722	35.2	790	62,341	7,262	20,729	30,773	29,256	34,787	64,044	45.6%	
301.3人・片働き(妻専業主婦)	1,739	33.8	522	54,316	6,055	10,829	32,320	30,750	33,545	64,295	37.1%	
302.3人・共働き(妻パート)	350	33.4	517	50,331	6,058	11,136	28,204	26,796	25,343	52,140	38.0%	
303.3人・共働き(妻フルタイム)	368	33.4	734	59,390	6,860	15,787	28,352	25,881	36,773	62,654	54.6%	
401.4人・片働き(妻専業主婦)	2,051	35.7	555	62,778	6,377	10,331	21,922	20,815	33,723	54,538	49.2%	
402.4人・共働き(妻パート)	611	35.4	558	63,582	7,056	11,057	23,138	22,143	32,755	54,898	47.6%	
403.4人・共働き(妻フルタイム)	379	37.2	794	70,067	7,808	14,291	18,105	17,406	69,797	87,203	66.5%	
501.5人以上・片働き(妻専業主婦)	885	36.8	582	73,691	7,502	9,454	20,872	17,936	36,389	54,325	50.3%	
502.5人以上・共働き(妻パート)	363	36.6	595	72,537	7,882	11,345	15,563	14,655	40,678	55,334	57.3%	
503.5人以上・共働き(妻フルタイム)	152	38.2	796	82,304	9,558	13,493	19,585	18,110	53,118	71,228	65.8%	

## 4.2. 預貯金・保険

消費支出以外の代表的な支出項目として、預貯金と保険の純増分を算出し、平均値と中央値を確認したところ、預貯金については妻フルタイム世帯とその他の世帯の間に大きな水準の開きがみられた（図表 12）。特に世帯人数3人以上の片働き世帯と妻パート世帯においては、平均値・中央値が相対的に低い水準となっている。また、預貯金純増がマイナスとなる世帯割合をしても、片働き世帯と妻パート世帯においては預貯金を取り崩している世帯が多いことが分かる。一方、保険については、保険金の受け取りが発生している世帯数が少ない（分析対象の98.7%の世帯で保険取金が0円）ため、預貯金ほど大きな水準の開きは見られない。

図表 12 世帯分類ごとにみた食料および住居への支出（実額）の詳細

	世帯数	世帯主 平均年齢	年間収入 (単位:万円)	預貯金純増(単位:円)		保険純増(単位:円)		預貯金純増<0 の世帯割合
				平均値	中央値	平均値	中央値	
201.2人・片働き(妻専業主婦)	876	35.9	549	12,241	21,649	19,225	15,287	40.3%
202.2人・共働き(妻パート)	476	37.1	610	15,135	22,989	20,127	15,918	39.5%
203.2人・共働き(妻フルタイム)	722	35.2	790	95,961	95,867	28,397	21,901	23.7%
301.3人・片働き(妻専業主婦)	1,739	33.8	522	10,776	15,571	21,784	17,561	42.7%
302.3人・共働き(妻パート)	350	33.4	517	-5,511	13,562	19,612	15,729	41.7%
303.3人・共働き(妻フルタイム)	368	33.4	734	82,668	78,845	29,574	23,863	20.4%
401.4人・片働き(妻専業主婦)	2,051	35.7	555	126	4,361	26,892	23,549	48.2%
402.4人・共働き(妻パート)	611	35.4	558	-5,365	3,032	28,725	23,382	48.8%
403.4人・共働き(妻フルタイム)	379	37.2	794	47,171	52,267	47,770	40,554	30.1%
501.5人以上・片働き(妻専業主婦)	885	36.8	582	1,142	10,945	33,452	30,543	44.9%
502.5人以上・共働き(妻パート)	363	36.6	595	-11,832	-3,373	28,981	26,037	52.1%
503.5人以上・共働き(妻フルタイム)	152	38.2	796	77,902	72,855	36,109	50,505	21.7%

※預貯金純増と保険純増は以下の式で算出した。

$$\text{預貯金純増} = Y191(\text{預貯金}) - Y025(\text{預貯金引出}) \quad \text{保険純増} = Y192(\text{保険掛金}) - Y026(\text{保険取金})$$

## 5. まとめ

本分析で確認できたことをまとめると、次の通りである。まず、夫婦を含む核家族世帯において、夫婦のみ世帯であれば妻がフルタイムで働いている割合が大きい。しかし、子供のいる世帯になると妻専業主婦世帯が増加する。しかし、子供の数が増えると妻パート世帯の割合の上昇により、共働き世帯割合が上昇する。世帯年収は妻専業主婦世帯と妻パート世帯はほぼ同程度の水準であるが、それらと妻フルタイム世帯の間には大きな差がある。収入面においては世帯人数による差がみられない一方で、支出面においては食料や教育など、子供の数が増えることでかかる費用が顕著に増加する項目があり、消費支出全体の差につながっている。そのしわ寄せが預貯金純増分に現れており、特に子育て中の妻専業主婦世帯と妻パート世帯では妻フルタイム世帯と比べて預貯金純増分の水準が低く、生活余力があまり無い様子が見られる。

こうした子育て中の生活余力が乏しい世帯の中には、子育てに重点を置くためにあえて専業主婦やパートタイマーを選んでいる家庭だけでなく、「妻もフルタイムで働きたいができない」といった家庭も含まれていると考えられる。実際に、都市部において待機児童問題が深刻であることが、そのような世帯が一定数いることの証左であろう。また、そうした労働意欲のある人たちを活かせないことは、労働力不足が顕著になりつつある日本社会にとっても損失であるだろう。待遇改善等を通じた保育士の確保により地域の保育需要を満たすことや、企業における社員の育児を応援する姿勢（制度面だけでなく、雰囲気も含めたもの）が望まれる。

## 参考文献

- [1] 井上貴博・矢田徹（2016）「匿名データを用いた夫婦共働き世帯における子あり世帯と夫婦のみ世帯の比較：横断研究」, 『SAS ユーザー総会 2016 論文集』, pp52-61.
- [2] 高浪洋平・舟尾暢男（2015）『改訂版 統計解析ソフト「SAS」』カットシステム.
- [3] SAS Institute Inc. SAS(R) 9.4 ODS Graphics: Procedures Guide, Sixth Edition; Available at : <http://documentation.sas.com/?docsetId=grstatproc&docsetTarget=titlepage.htm&docsetVersion=9.4&locale=ja> (Accessed at July 4, 2018)

## 付録 1 規定課題用 SAS プログラム

```
/* ライブラリ・出力用PDFファイルの指定 */
%GLOBAL IN_LIB OUT_PDF;
%LET IN_LIB = フォルダパス; /* ←新疑似マイクロデータの
SASデータセットが入っているフォルダパスを入力 */
%LET OUT_PDF = フォルダパス ; /* ←規定課題の表出力PDF
を出力するフォルダパスを入力 */
```

```
/* ライブラリ割り当て */
```

```
libname micro "&IN_LIB" access = readonly ;
```

```
/* フォーマット作成 */
```

```
proc format;
  value X08f
    1 = '1:男'
    2 = '2:女'
    ;
  value X11f
    1 = '1:単身世帯'
    2 = '2:夫婦のみ世帯'
    3 = '3:二世世代世帯'
    4 = '4:二世代（ひとり親）世帯'
    5 = '5:三世世代世帯'
    6 = '6:その他の世帯'
    7 = '7:全世帯'
    ;
```

```
run;
```

```
/*-----*
規定課題 1 : 家族分類毎、世帯主の性別別および全世帯の年
間収入（万円）の平均値・標準偏差と世帯数
-----*/
```

```
/* 出力用データセットの作成 */
```

```
proc means data = micro.zensho2004gijimicro
  vardef = wdf
  mean stddev sumwgt
  noprint
  ;
  class X08 X11;
  var Y001;
  weight weight;
  output out = work.income_1
    mean = mean stddev = stddev sumwgt = sumwgt;
run;
```

```
data work.income_2;
  set work.income_1;
  if X08 = . then do;
    X08 = 3;
  end;
  if X11 = . then do;
    X11 = 7;
```

```
end;
run;
```

```
proc sort data = work.income_2 out = work.income_3;
  by X11 X08;
run;
```

```
data work.income_4;
  merge work.income_3(where = (X08 = 1) rename = (mean
= mean_1 stddev = stddev_1 sumwgt = sumwgt_1))
    work.income_3(where = (X08 = 2) rename = (mean
= mean_2 stddev = stddev_2 sumwgt = sumwgt_2))
    work.income_3(where = (X08 = 3) rename = (mean
= mean_3 stddev = stddev_3 sumwgt = sumwgt_3))
    ;
  by X11;
  keep X11 mean_1 - mean_3 stddev_1 - stddev_3 sumwgt_1
- sumwgt_3;
run;
```

```
/* 出力は規定課題2のデータ後にまとめて行う */
```

```
/*-----*
規定課題 2 : 世帯主の性別別の年間収入（万円）のジニ係数
を小数点第2位まで
-----*/
```

```
/* 計算用データセットの作成：データセットを縦積みし、計
算用のグループフラグ(culc_flg)を作る */
```

```
data work.gini_1;
  set micro.zensho2004gijimicro(keep = X08 X11 Y001
weight in = in_1)
    micro.zensho2004gijimicro(keep = X08 X11 Y001
weight in = in_2)
    micro.zensho2004gijimicro(keep = X08 X11 Y001
weight in = in_3)
    micro.zensho2004gijimicro(keep = X08 X11 Y001
weight in = in_4);
```

```
length culc_flg $2;
```

```
if in_1 eq 1 then do;
  X08_2 = X08; X11_2 = X11;
end;
else if in_2 eq 1 then do;
  X08_2 = 3; X11_2 = X11;
end;
else if in_3 eq 1 then do;
  X08_2 = X08; X11_2 = 7;
end;
else if in_4 eq 1 then do;
  X08_2 = 3; X11_2 = 7;
end;
```

```

    culc_flg = cats(X08_2, X11_2);
run;

/* freqプロシジャでby変数に指定するためソート */

proc sort data = work.gini_1 out = work.gini_2;
    by culc_flg;
run;

/* freqプロシジャでグループごとの相対度数のデータを吐き出す */

proc freq data = work.gini_2 noprint;
    weight weight;
    by culc_flg;
    tables Y001 / out= work.gini_3 outcum;
run;

/* 収入階級ごとの収入×度数の計算 */

data work.gini_4;
    set work.gini_3;

    ranksum = Y001 * count;    /* 収入×度数 */
run;

/* 計算用グループごとの収入の合計 (Σ (収入×度数)) を持ったデータセット (gini_4a) を作る */

proc means data = work.gini_4 sum nway noprint;
    class culc_flg;
    var ranksum ;
    output out = gini_4a sum = totalmoney;
run;

/* gini_4aをgini_4にマージし、ジニ係数の算出に必要な数値を算出 (収入の相対度数など) */

data work.gini_5 ;
    merge
        work.gini_4
        work.gini_4a(keep = culc_flg totalmoney);
    by culc_flg;

    rankfreq = ranksum/totalmoney; /* その収入の相対度数 */
    if first.culc_flg then cumrankfreq = 0;
    cumrankfreq + rankfreq; /* その収入額の累積相対度数 */

    dai = ( lag1(cumrankfreq) + cumrankfreq ) * ( percent / 100 ) / 2 ; /* 台形の面積 */

    if first.culc_flg then dai = cumrankfreq * ( percent / 100 ) / 2;

    if first.culc_flg then daisum = 0;
    daisum + dai; /* 台形の面積の和 */

```

```

run ;

/* ジニ係数を計算・後のデータセットの転置・出力用にX08とX11を復活させる*/

data work.gini_6;
    format X08 X11 culc_flg daisum ginicoef;
    set work.gini_5(keep = culc_flg daisum);
    by culc_flg;

    format ginicoef 8.2;

    ginicoef = (0.5 - daisum) * 2;

    X08 = input(substr(culc_flg, 1, 1), 8.);
    X11 = input(substr(culc_flg, 2, 1), 8.);

    keep culc_flg ginicoef X08 X11;

    if last.culc_flg then output;

run;

/* 出来上がったデータセットの転置*/
data work.gini_result;
    merge work.gini_6(where = (X08 = 1) rename = (ginicoef = ginicoef_1))
        work.gini_6(where = (X08 = 2) rename = (ginicoef = ginicoef_2))
        work.gini_6(where = (X08 = 3) rename = (ginicoef = ginicoef_3));
    by X11;
    keep X11 ginicoef_1 - ginicoef_3;
run;

/*-----
図表の出力
-----*/

options nodate nonumber ;

ods pdf file = "&OUT_PDF%kitei_kadai.pdf";

title ' 付録2 規定課題図表';

ods layout start columns = 1 rows = 2;

/* 規定課題1 */

ods region;

title1 ' 【規定課題1】';
title2 ' 家族分類・世帯主の性別毎の年間収入の平均値・標準偏差と世帯数';

data _null_;

    set work.income_4 end = EOF;
    if _N_ eq 1 then do;
        dcl odsout obj();
        obj.table_start();

```

```

/* 1行目 */
obj.row_start();
obj.format_cell(data:'', row_span:3);
obj.format_cell(data:'世帯主の性別', column_span:2);
obj.format_cell(data:'全世帯', row_span:2, vjust: 'M');
obj.format_cell(data:'世帯主の性別', column_span:2);
obj.format_cell(data:'全世帯', row_span:2, vjust: 'M');
obj.format_cell(data:'世帯主の性別', column_span:2);
obj.format_cell(data:'全世帯', row_span:2, vjust: 'M');
obj.row_end();

/* 2行目 */
obj.row_start();
obj.format_cell(data:'1:男');
obj.format_cell(data:'2:女');
obj.format_cell(data:'1:男');
obj.format_cell(data:'2:女');
obj.format_cell(data:'1:男');
obj.format_cell(data:'2:女');
obj.row_end();

/* 3行目 */
obj.row_start();
obj.format_cell(data:'平均値', column_span:3);
obj.format_cell(data:'標準偏差', column_span:3);
obj.format_cell(data:'世帯数', column_span:3);
obj.row_end();

/* 4行目 */
obj.row_start();
obj.format_cell(data:'家族分類');
obj.format_cell(data:'', column_span:3);
obj.format_cell(data:'', column_span:3);
obj.format_cell(data:'', column_span:3);
obj.row_end();

end;

/* データの中身出力 */

obj.row_start();
obj.format_cell(data:X11, format:'X11f.',
style_attr:'just = L background = white');
obj.format_cell(data:mean_1, format:'8.',
style_attr:'just = R background = very light yellow');
obj.format_cell(data:mean_2, format:'8.',
style_attr:'just = R background = very light yellow');
obj.format_cell(data:mean_3, format:'8.',
style_attr:'just = R background = very light yellow');
obj.format_cell(data:stddev_1, format:'8.',
style_attr:'just = R background = very light yellow');
obj.format_cell(data:stddev_2, format:'8.',
style_attr:'just = R background = very light yellow');
obj.format_cell(data:stddev_3, format:'8.',
style_attr:'just = R background = very light yellow');
obj.format_cell(data:sumwgt_1, format:'comma11.',
style_attr:'just = R background = very light yellow');
obj.format_cell(data:sumwgt_2, format:'comma11.',
style_attr:'just = R background = very light yellow');
obj.format_cell(data:sumwgt_3, format:'comma11.',
style_attr:'just = R background = very light yellow');
obj.row_end();

if EOF then obj.table_end();

```

```

run;

/* 規定課題2 */
ods region;

title1 '【規定課題2】';
title2 '家族分類・世帯主の性別毎の年間収入のジニ係数';

data _null_;

set work.gini_result end = EOF;
if _N_ eq 1 then do;
dcl odsout obj;
obj.table_start();

/* 1行目 */
obj.row_start();
obj.format_cell(data:'', row_span:3);
obj.format_cell(data:'年間収入のジニ係数',
column_span:3);
obj.row_end();

/* 2行目 */
obj.row_start();
obj.format_cell(data:'世帯主の性別',
column_span:2);
obj.format_cell(data:'全世帯', row_span:2);
obj.row_end();

/* 3行目 */
obj.row_start();
obj.format_cell(data:'1:男');
obj.format_cell(data:'2:女');
obj.row_end();

/* 4行目 */
obj.row_start();
obj.format_cell(data:'家族分類');
obj.format_cell(data:'', column_span:3);
obj.row_end();

end;

/* データの中身出力 */

obj.row_start();
obj.format_cell(data:X11, format:'X11f.',
style_attr:'just = L background = white');
obj.format_cell(data:ginicoef_1, format:'8.2',
style_attr:'just = R background = very light yellow');
obj.format_cell(data:ginicoef_2, format:'8.2',
style_attr:'just = R background = very light yellow');
obj.format_cell(data:ginicoef_3, format:'8.2',
style_attr:'just = R background = very light yellow');
obj.row_end();

if EOF then obj.table_end();
run;

title '';
ods layout end;
ods pdf close;

```

## 付録2 規定課題図表

### 【規定課題1】

家族分類・世帯主の性別毎の年間収入の平均値・標準偏差と世帯数

家族分類	世帯主の性別		全世帯	世帯主の性別		全世帯	世帯主の性別		全世帯
	1:男	2:女		1:男	2:女		1:男	2:女	
	平均値			標準偏差			世帯数		
1:単身世帯	416	266	338	267	165	232	3,187,182	3,432,001	6,619,183
2:夫婦のみ世帯	557	539	557	361	331	360	7,600,840	143,504	7,744,344
3:二世帯世帯	733	686	732	373	366	373	11,709,238	147,052	11,856,290
4:二世帯(ひとり親)世帯	689	459	558	402	326	378	1,003,841	1,324,421	2,328,262
5:三世帯世帯	923	677	904	459	421	461	2,545,855	210,980	2,756,835
6:その他の世帯	778	576	724	456	429	458	428,659	153,734	582,393
全世帯	662	356	610	396	281	396	26,475,615	5,411,692	31,887,307

### 【規定課題2】

家族分類・世帯主の性別毎の年間収入のジニ係数

家族分類	年間収入のジニ係数		
	世帯主の性別		全世帯
	1:男	2:女	
1:単身世帯	0.31	0.32	0.34
2:夫婦のみ世帯	0.31	0.31	0.31
3:二世帯世帯	0.26	0.28	0.26
4:二世帯(ひとり親)世帯	0.30	0.36	0.35
5:三世帯世帯	0.26	0.34	0.27
6:その他の世帯	0.30	0.37	0.33
全世帯	0.31	0.38	0.33

## 付録3 自由課題用 SAS プログラム

```

/* ライブラリ割り当て */
libname micro 'フォルダパス' access = readonly ;

/* 分析対象(n=8,972)の抽出 */
data kakukazoku;
  set micro.zensho2004gijimicro;
  where X02 = 1 and X08 = 1 and X09 <= 10 and
X11 in (2, 3) and Y018 = 0;

  length hatarakite_flg $3;
  if missing(Y006) eq 0 then do;
    if Y006 > 0 then do; Y006_flg = '1'; end;
    else if Y006 = 0 then do; Y006_flg = '0'; end;
    else do; Y006_flg = '9'; end;
  end;
  else do; Y006_flg = '9'; end;
end;
if missing(Y007) eq 0 then do;
  if Y007 > 0 then do; Y007_flg = '1'; end;
  else if Y007 = 0 then do; Y007_flg = '0'; end;
  else do; Y007_flg = '9'; end;
end;
else do; Y007_flg = '9'; end;
end;
if missing(Y008) eq 0 then do;
  if Y008 > 0 then do; Y008_flg = '1'; end;
  else if Y008 = 0 then do; Y008_flg = '0'; end;
  else do; Y008_flg = '9'; end;
end;
else do; Y008_flg = '9'; end;
end;
hatarakite_flg = cats(Y006_flg, Y007_flg, Y008_flg);

if X04 = 1 and hatarakite_flg = '100' then do;
  h_valid_flg = 1; tomo_flg = 0; end;
else if X04 = 2 and hatarakite_flg = '110' then do;
  h_valid_flg = 1; tomo_flg = 1; end;
else do; h_valid_flg = 0; tomo_flg = .; end;

if X11 = 3 then do;
  if X12 = 2 then do; c_valid_flg = 1; end;
  else do; c_valid_flg = 0; end;
end;
else do; c_valid_flg = 1; end;
end;

if h_valid_flg = 1 and c_valid_flg = 1 then output;
run;

/* 片働き・共働き（フルタイム/パート）フラグ・変数作成
（一部抜粋）*/
data kakukazoku_2;
  set kakukazoku;
  attrib
setai_bunrui length = $2 label = '妻の就業形態別世帯分
類'

```

```

cld_setai_bunrui length = $3 label = '世帯人員・妻の就
業形態別世帯分類';

if tomo_flg = 1 then do;
  fulltime = ifn(Y007*12 <= 1410000, 0, 1);
end;
else if tomo_flg = 0 then do; fulltime = 0; end;
else do; fulltime = .; end;

setai_bunrui = cats(tomo_flg, fulltime);
cld_setai_bunrui = cats(X03, tomo_flg, fulltime);
run;

/* フォーマット作成 */
proc format;
  value $ setai_bunrui_f
    '00' = '01_片働き(妻専業主婦)'
    '10' = '02_共働き(妻パート)'
    '11' = '03_共働き(妻フルタイム)'
  value $ cld_setai_bunrui_f
    '200' = '201_2人・片働き(妻専業主婦)'
    /* (中略) */
    '511' = '503_5人以上・共働き(妻フルタイム)';
run;

/* 箱ひげ図(図表8)の再現 */
options orientation = landscape;
ods rtf file = 'フォルダパス\hakohige1.rtf' ;
%SGANNO;
data hakohigeanno;
  %SGTEXT(x1space = "graphpercent", y1space =
"graphpercent", x1 = 2, y1 = 7, label = "(注)「ひげ」
は箱の上端・下端から四分位範囲×1.5までの範囲で最も離れた
観測値まで伸びている。[2][3]", textcolor = "black",
textsize = 10, width = 100, anchor = "left");
  %SGTEXT(x1space = "graphpercent", y1space =
"graphpercent", x1 = 2, y1 = 3, label = "(注)この箱ひ
げ図では外れ値の出力を抑制している。", textcolor =
"black", textsize = 10, width = 100, anchor = "left");
run;
ods graphics /height = 400px width = 900px attrpriority
= none;
proc sgplot data = kakukazoku_2 sganno = hakohigeanno
pad = (bottom = 40pt);
  styleattrs datacolors = (lightblue lightred
lightgreen) datasymbols = (circle square diamond);
  hbox Y001 / category = cld_setai_bunrui group =
setai_bunrui nooutliers lineattrs = (color = black)
whiskerattrs = (color = black) meanattrs = (color =
black) medianattrs = (color = black);
  xaxis values=(0 to 1800 by 200) label='世帯収入(単位
万円)';
  format cld_setai_bunrui $cld_setai_bunrui_f.
setai_bunrui $setai_bunrui_f.;
run;
ods rtf close;

```